
'**************************************************
'Used to calibrate and adjust gauge mode PMMs.  Called by
'the master test macro "PMMCalSled_Master".

'******************************************************************************

Function FlukeCalSledGauge(iT, iL, iC, iP, cTest, cConfig)
Dim C1_Removed_Data()
Dim Reference_Data()
Dim C1_Corrected_Data()
Dim res(4,7)

C0 = 0
C1 = 0
zOffset = 0

'<currentteststep 1010 - record AF coefs.>
If cCOMPASS.CurrentTestStep = 1010 Then ' Test Definition data files were created for each DUT.
     cDebug.LogStatus "Capture As Found Coeffs"
     
     For i = 1 To cConfig.DUTPrs.Count 
[bookmark: _GoBack]           cDebug.LogStatus "DUT: " & i
           
           ' Read A/F C0 from PMM; write to calcoef1:
           C0 = cConfig.DUTPrs(CInt(i)).IoSendCommand("OFFSET_SET?", False)
           cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient1 = C0
                cDebug.LogStatus "Command OFFSET_SET?: " & C0
                cDebug.LogStatus "CalibrationCoefficient1: " & cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient1
           
           ' Read A/F C1 from DUT; write to calcoef2
           C1 = cConfig.DUTPrs(CInt(i)).IoSendCommand("GAIN_SET?", False)
           cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient2 = C1
                cDebug.LogStatus "Command GAIN_SET?: " & C1
                cDebug.LogStatus "CalibrationCoefficient2: " & cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient2
           
           ' Read A/F zOffset from PMM, write to calcoef3:
           zOffset = cConfig.DUTPrs(CInt(i)).IoSendCommand("UCOEF_SET[0]?", False)
           cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient3 = zOffset
                cDebug.LogStatus "Command UCOEF_SET[0]?: " & zOffset
                cDebug.LogStatus "CalibrationCoefficient3: " & cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient3
     
           calDate = cConfig.DUTPrs(CInt(i)).IoSendCommand("CAL_DATE?", False)
           cCOMPASS.DataCollection(i).DUT.LastCalDate = calDate
                cDebug.LogStatus "Command CAL_DATE?: " & calDate
                cDebug.LogStatus "LastCalDate: " & cCOMPASS.DataCollection(i).DUT.LastCalDate

     Next
     cdebug.LogStatus "End of CurrentTestStep 1010"


'<currenteststep 1100 - zero the PMM>
ElseIf cCOMPASS.CurrentTestStep = 1100 And iC = 1 Then  ' New test pressure cycle and the 1st one
     cDebug.LogStatus "Starting step 1100"
     
     	'DaqType mode = 0: manual Set device
	     DAQ = cCOMPASS.cConfig.SetPrs(1).RangeMain.GetParent.DaqType
	     cdebug.LogStatus "Data Acquisition Type for SetPrs is: " & DAQ
     
	     If DAQ <> 0 Then     
	     	cCOMPASS.StatusDisplay "Venting the pressure..."
	     	cCOMPASS.cConfig.SetPrs(1).IoSetOutput 0,0,1
	     	cDebug.LogStatus "Sent ioSetOutput 0,0,1 to vent"
	     	TimeDelay 2
     
	     ElseIf DAQ = 0 Then
		     msg = msgbox("Vent the Pressure, Press OK when fully vented.",0+64+4096,"Manual Vent")
		     cCOMPASS.StatusDisplay "10s Dwell for Stabilization..."
		     TimeDelay 7
	          
	     End If
     
     For i = 1 To cConfig.DUTPrs.Count
     
	     'reset the Zoffset to "0.000", however this does not fully "zero" out the PMM as the C0 coef 
	     'Is still being applied...
	     cCOMPASS.StatusDisplay "Removing the zero offset..."
	     TimeDelay 2
	     a = cCOMPASS.cConfig.DUTPrs(i).IoSendCommand("UCOEF_SET[0] 0",True)
	     TimeDelay 2
	     cDebug.LogStatus "sent cmd to zero Zoffset"
	       
		 b = cCOMPASS.cConfig.DUTPrs(i).IoSendCommand("VAL?",False)	
		 cDebug.LogStatus "VAL? return string is: " & b
	 
		 'To do math on the pressure value it must be in a Data varient type.  Must first extract the 
		 'number part from the string and then convert to a double precision data variant.
		 PMMValue = qextract(b,0,1,",")
		 cDebug.LogStatus "the PMMValue is: " & PMMValue
		 
		 'convert the string to a data varient:
		 c = CDbl(PMMValue)
		 cDebug.LogStatus "the data type value for PPMValue is: " & c
 	
		 newOffset = c * -1
		 cDebug.LogStatus "the newOffset value is: " & newOffset
		
		 'finish the zeroing process...	 
		 cmd = "UCOEF_SET[0] " & newOffset
		 e = cCOMPASS.cConfig.DUTPrs(i).IoSendCommand(CStr(cmd), True)
		 cDebug.LogStatus "offset command sent to the PMM: " & cmd
	 
	 	 'verify new zoffset correctly received by PMM, save it to DUTRaw3:
		 zOffset = cConfig.DUTPrs(i).IoSendCommand("UCOEF_SET[0]?", False)
		 cCOMPASS.DataCollection(i).DUTRaw3.CalibrationCoefficient3 = zOffset
	 
     Next
     cDebug.LogStatus "End of CurrentTestStep 1100"
     
     

'Continue with pressure cycle 1 (as-found data)

'<currenteststep 1150 - pressure cycle 1 is done, run polyfit>
ElseIf cCOMPASS.CurrentTestStep = 1150 And iC = 1 Then 
     cCOMPASS.StatusDisplay "Calculating Adjustment..." 
           cDebug.LogStatus "First Cycle is complete. Starting CurrentTestStep 1150 - run PolyFit"
           cDebug.LogStatus "#Data files: " & cCOMPASS.DataCollection.Count  

     For i = 1 To cConfig.DUTPrs.Count 
           pressurePoints = cCOMPASS.DataCollection(i).NumberofPressurePoints
                cDebug.LogStatus "DUT: " & i & " - Pressure Points: " & pressurePoints
                
           Redim Reference_Data(pressurePoints-1)
           Redim C1_Removed_Data(pressurePoints-1)
           Redim C1_Corrected_Data(pressurePoints-1)

           'populate PMM A/F coefficients from memory:
           C0 = cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient1
           C1 = cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient2
           zOffset = cCOMPASS.DataCollection(i).DUTRaw3.CalibrationCoefficient3
                cDebug.LogStatus "C0: " & C0  
                cDebug.LogStatus "C1: " & C1
                cDebug.LogStatus "zOffset: " & zOffset

           dutUnit = cConfig.DUTPrs(i).RangeMain.UnitFinal
                cDebug.LogStatus "DUT UnitFinal: " & dutUnit
                cDebug.LogStatus "DUT UnitFinalText: " & cConfig.DUTPrs(i).RangeMain.UnitFinalText
           refUnit = cConfig.RefPrs(i).RangeMain.UnitFinal
                cDebug.LogStatus "Ref UnitFinal: " & refUnit
                cDebug.LogStatus "Ref UnitFinal: " & cConfig.RefPrs(i).RangeMain.UnitFinalText
           
           ' Iterate through each pressure point
           For j = 1 To pressurePoints
                ix = CInt(j)-1
                     cDebug.LogStatus "FlukeCalSledCalibrationAdjust DUTPressurePoint(" & j & ")"
                DUTPressure = cCOMPASS.DataCollection(i).DataPointRef(1, 1, 1, CInt(j)).DUTPressure
                     cDebug.LogStatus "FlukeCalSledCalibrationAdjust DUTPressurePoint: " & DUTPressure
                DUTPsiPres = cCOMPASS.UnitConversion(CDbl(DUTPressure), 9, CInt(dutUnit), 0) 
                     cDebug.LogStatus "FlukeCalSledCalibrationAdjust DUTPressurePoint (psi): " & DUTPsiPres
                
                refPres = cCOMPASS.DataCollection(i).DataPointRef(1, 1, 1, CInt(j)).RefPressure
                     cDebug.LogStatus "FlukeCalSledCalibrationAdjust ReferencePressurePoint: " & refPres
                Reference_Data(ix) = cCOMPASS.UnitConversion(CDbl(refPres), 9, CInt(refUnit), 0) 
                     cDebug.LogStatus "FlukeCalSledCalibrationAdjust ReferencePressurePoint (psi): " & Reference_Data(ix)
                
                zOff_Removed = CDbl(DUTPsiPres) - CDbl(zOffset)
                     cDebug.LogStatus "FlukeCalSledCalibrationAdjust zOff_Removed: " & zOff_Removed
                C0_Removed = zOff_Removed - C0
                     cDebug.LogStatus "FlukeCalSledCalibrationAdjust C0_Removed: " & C0_Removed
                C1_Removed_Data(ix) = C0_Removed / C1
                     cDebug.LogStatus "FlukeCalSledCalibrationAdjust C1_Removed_Data: " & C1_Removed_Data(ix)
           Next
           
           Call Poly_Fit(Reference_Data, C1_Removed_Data, res, 1)
           slope= res(0,1)
                cDebug.LogStatus "FlukeCalSledCalibrationAdjust slope: " & slope
           If slope = 0 Then
                new_C1 = 0
           Else
                new_C1 = 1 / slope
           End If
                cDebug.LogStatus "FlukeCalSledCalibrationAdjust new_C1: " & new_C1
           
           'Calculate the C1_corrected for each point:
           For j = 0 To pressurePoints - 1
                C1_Corrected_Data(CInt(j)) = C1_Removed_Data(CInt(j)) * new_C1
                cDebug.LogStatus "FlukeCalSledCalibrationAdjust C1_Corrected_Data(" & j & "): " & C1_Corrected_Data(CInt(j))
           Next
           
           Call Poly_Fit(Reference_Data, C1_Corrected_Data, res, 1)
           new_C0 = 0 - res(0,0)
           cDebug.LogStatus "FlukeCalSledCalibrationAdjust new_C0: " & new_C0
           
           'Save new coefs to memory AND write to the PMM:
           cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient4 = new_C0
           cmd = "OFFSET_SET " & new_C0
           Call cConfig.DUTPrs(CInt(i)).IoSendCommand(CStr(cmd), True)
                cDebug.LogStatus "CalibrationCoefficient4: " & cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient4
                cDebug.LogStatus "Sent command: " & cmd
           
           cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient5 = new_C1
           cmd = "GAIN_SET " & new_C1
           Call cConfig.DUTPrs(CInt(i)).IoSendCommand(CStr(cmd), True)
                cDebug.LogStatus "CalibrationCoefficient5: " & cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient5
                cDebug.LogStatus "Sent command: " & cmd
           
           'Determine new date of calibration, save to memory AND write to PMM:
           new_date = FlukeCalSled_Date
           cCOMPASS.DataCollection(i).DUT.CalDueDate = CDate(new_date)
           cmd = "CAL_DATE " & new_date
           Call cConfig.DUTPrs(CInt(i)).IoSendCommand(CStr(cmd), True)
                cDebug.LogStatus "CalDueDate: " & cCOMPASS.DataCollection(i).DUT.CalDueDate
                cDebug.LogStatus "Sent command: " & cmd
     
           'Reset Zoffset to zero:
           cmd = "UCOEF_SET[0] 0" 
           Call cConfig.DUTPrs(CInt(i)).IoSendCommand(CStr(cmd), True)
                cDebug.LogStatus "Sent command: " & cmd
           
     Next
     
     Set obj = Nothing
     cdebug.LogStatus "Finished with CurrentTestStep 1150"


'<currentteststep 2000 - test is complete, AL data has been collected. Save the coefs>
' Test complete. The post test options have not displayed and the user notes have not been entered.
ElseIf cCOMPASS.CurrentTestStep = 2000 Then  

     cDebug.LogStatus "Test complete.  CurrentTestStep 2000.  Get final zOffset"
     
'Capture final zOffset - should be zero
   For i = 1 To cConfig.DUTPrs.Count 
          cDebug.LogStatus "DUT: " & i
           ' Read zOffset from DUT
          zOffset = cConfig.DUTPrs(CInt(i)).IoSendCommand("UCOEF_SET[0]?", False)
          cDebug.LogStatus "Command UCOEF_SET[0]?: " & zOffset
           ' Write the final zOffset value
          cCOMPASS.DataCollection(i).DUT.CalibrationCoefficient6 = zOffset
    Next 

'OPTIONAL GUARD BANDING FEATURE:  This block of code can be commented out if guardbanding evaluation is not desired.
		TOLFACT = .50   '50% gaurdband.  User definable.
		
		For i = 1 To cCOMPASS.DataCollection(1).NumberofPressurePoints
			For j = 1 To cConfig.DUTPrs.Count 
			     Ref = cCOMPASS.DataCollection(j).DataPointRef(1,1,2,CInt(i)).RefPressure
			     DUT = cCOMPASS.DataCollection(j).DataPointRef(1,1,2,CInt(i)).DUTPressure
			     Tol  = cCOMPASS.DataCollection(j).DataPointRef(1,1,2,CInt(i)).Tolerance  
				 ST = cCOMPASS.DataCollection(j).DataPointRef(1,1,2,CInt(i)).Status 
			
			     If abs(DUT-Ref) > Tol * TOLFACT Then 
			           FAIL = True
			           cCOMPASS.DataCollection(j).DataPointRef(1,1,2,CInt(i)).Status = ST & "T" 
			           'Generate Failure message for user
			            msgbox "The As Left results did not fall inside the guardbanded tolerance. Click OK to continue."
			            Exit For 
					End If 
			Next
			If fail = True Then Exit For
'END OF GUARD BANDING CODE.


	Next
     msg = "The test is complete.  The coefficients are active in module memory but "
     msg = msg & "have not been written to permanent storage. Would you like to activate " 
     msg = msg & "the calibration?"
     
     If msgbox(msg,vbquestion + vbYesNo + vbSystemModal,"Activate Calibration") = vbNO Then 
           msg = "Recycling the power of the module will "
           msg = msg & "remove the coefficients from the memory and reset the module."
           msgbox msg,vbSystemModal,"Activate Calibration"
           Exit Function         
     Else
           For i = 1 To cConfig.DUTPrs.Count 
                cmd = "CAL_STORE"
		 TimeDelay 2
                Call cConfig.DUTPrs(CInt(i)).IoSendCommand(CStr(cmd), True)
                cDebug.LogStatus "Sent command: " & cmd
           Next 
     End If

End If
End Function

